DYNAMICAL GRAVASTARS

Stephen L Adler, JAS

EVENT HORIZON TELESCOPE (EHT)

SEES "LIGHT SPHERE" $r = 3M$

WHAT IS INSIDE ?

- BLACK HOLE $r = 2M$?
- COMPACT OBJECT WITH NO HORIZON $\underbrace{(ECO, UCO)}_{\text{GRAVASTAR}}$?

PROBLEMS WITH HORIZONS

PARADOXES:

- INFORMATION
- SINGULARITY THEOREMS
- TRILLIONS OF CAUSALLY DISCONNECTED REGIONS

BIRKHOFF THEOREM:

$g_{00} \approx 1 - \frac{2M}{r}$ IN EXTERIOR

$\nRightarrow$ $g_{00} = 0$ AT $2M$

$g_{00} < 0$ AT $r < 2M$

ANY COMPACT MASS HAS SAME VACUUM METRIC

COMPACT OBJECTS

- EXOTIC COMPACT OBJECT (ECO) REVIEW:

  CARDOSO + PANI arXiv: 1904.05363

- GRAVITY VACUUM STAR — GRAVASTAR

  BASED ON GLINER (1965) PROPOSAL OF A PHASE TRANSITION

  ORDINARY MATTER EQ OF STATE → COSMOLOGICAL CONSTANT "VACUUM" EQ OF STATE

  $p + \rho = 0$

- MAZUR + MOTTOLA arXiv: gr-qc/0109035

  ASSUME $\rho > 0$, p JUMPS TO $p < 0$

  SPECIFIES RADII WHERE TRANSITIONS OCCUR

- DYNAMICAL GRAVASTARS

ADLER arXiv: 2209.02537 (Phys. Rev. D 106, 104061 (2022))

JUST ANOTHER KIND OF RELATIVISTIC STAR

WHITE DWARF → NEUTRON STAR → GRAVASTAR

- TOLMAN-OPPENHEIMER-VOLKOFF EQUATIONS

+ EQUATION OF STATE

NOTHING ELSE!

- TOV ⇒ p CONTINUOUS

SO DENSITY ρ MUST JUMP

$\rho \to \rho < 0$ TO GET $p + \rho = 0$

- I TAKE $p + \rho = \beta > 0$ SMALL

IMPLICATIONS FOR ENERGY CONDITIONS

- NULL ENERGY CONDITION $\rho + p \geq 0$ (PENROSE SINGULARITY THM)

  OBEYED BY $\rho + p = \beta > 0$

- STRONG ENERGY CONDITION $\rho + p \geq 0$, $\rho + 3p \geq 0$ (HAWKING SINGULARITY THM)

  OBEYED BY $\rho + 3p = \beta + 2p > 0$

- THE DOMINANT ENERGY CONDITION $\rho \geq |p|$

  WEAK ENERGY CONDITION $\begin{cases} \rho \geq 0 \\ \rho + p \geq 0 \end{cases}$

  ARE VIOLATED

THE CONTINUITY ARGUMENT

$$\frac{dF(\lambda)}{d\lambda} = G(\lambda) \qquad \lambda_A \leq \lambda \leq \lambda_B$$

$$|G(\lambda)| \leq K < \infty \Rightarrow F(\lambda) \text{ CONTINUOUS}$$

PROOF:

$$|F(\lambda_0+\delta) - F(\lambda_0-\delta)| \leq \int_{\lambda_0-\delta}^{\lambda_0+\delta} d\lambda\, |G(\lambda)| \leq 2\delta K$$

$$\Rightarrow \text{ AS } \delta \to 0 \qquad F(\lambda_0^+) = F(\lambda_0^-)$$

CONTINUITY

$\Rightarrow$ **G CAN JUMP, F MUST BE CONTINUOUS**

TOV EQUATIONS

- LINE ELEMENT

$$ds^2 = e^{\nu(r)} dt^2 - e^{\lambda(r)} dr^2 - r^2 d\Omega^2$$

- TOV EQUATIONS

$$\frac{dm(r)}{dr} = 4\pi r^2 \rho(r) \quad , \quad e^{-\lambda(r)} = 1 - \frac{2m(r)}{r}$$

$$\frac{d\nu(r)}{dr} = \frac{N_\nu}{1 - \frac{2m(r)}{r}} \quad , \quad N_\nu = \frac{2}{r^2}\left[m(r) + 4\pi r^3 p(r)\right]$$

$$\frac{dp(r)}{dr} = -\frac{\rho(r) + p(r)}{2} \frac{d\nu(r)}{dr} \quad , \quad \text{WHEN } \rho + p = 0 \quad \frac{dp}{dr} = 0$$

- INITIAL CONDITIONS

$m(0) = 0$ $\quad$ $p(0) = 1$ $\quad$ $\nu(0) = $ nuInit picked to make $\nu(\infty) = 0$

"SHOOTING" METHOD TO TUNE $\nu(0) = \text{nuinit}$

DEFINE

$$M(r) = \left[1 - e^{\nu(r)}\right] \cdot \frac{r}{2}$$

ie

$$e^{\nu(r)} = 1 - \frac{2M(r)}{r}$$

SLOPE OF $M(r)$ AS $r \to \infty$ IS

$$S = \frac{1 - e^{\nu(\infty)}}{2}, \quad M = S\,r$$

TUNE nuinit TO GET $S = 0$

PROGRAM PARAMETERS

- GITLAB PROGRAMS

  $\beta = .1, .01, .001$

  COSMOLOGICAL CONSTANT Λ INCLUDED

  ACCURACY = 13 PRECISION = 13

  (NEED ≥ 5)

  SIGMOIDAL (SMOOTHED) JUMP

- WOLFRAM COMMUNITY DEMO NOTEBOOK

  $\beta = .01$ ONLY

  $\Lambda = 0$

CITE THIS NOTEBOOK: Dynamical Gravastars by *Stephen Adler*. Wolfram Community JAN 31 2023.
ORIGINAL ARTICLE: Stephen L. Adler, Dynamical Gravastars, Phys. Rev. D **106**, 104061 (2022), https://doi.org/10.1103/PhysRevD.106.104061. arXiv:2209.02537.
NOTEBOOKS: https://gitlab.com/stephenadler/Gravastar

Article Abstract: *We combine the ideas of a Weyl scaling invariant dark energy action, which eliminates black hole horizons, with the "gravastar" idea of a jump in the hole interior from a normal matter equation of state to an equation of state where pressure plus density approximately sum to zero. Using the Tolman-Oppenheimer-Volkoff equation, which requires continuous pressure, we present Mathematica notebooks in which the structure of the gravastar is entirely governed by the action and the equation of state, with the radii where structural changes occur emerging from the dynamics, rather than being specified in advance. The notebooks work even with zero cosmological constant, but when the cosmological constant is nonzero, there is a very small black hole "wind" that we calculate by a relativistic extension of standard pressure driven isothermal stellar wind theory.*

```
(Debug) In[ ]:= (*Princeton talk version of demo notebook*)
```

I wish to thank Fethi Ramazanoglu for teaching me to use Mathematica for relativity calculations in the course of our working on our joint paper Int . J. Mod. Phys. D **24**, 1550011 (2015), arXiv: 1308.1448. This explored a mechanism for eliminating horizons by using a modified cosmological constant action. My Dynamical Gravastars paper combined this idea with an alternative mechanism involving a jump in the interior matter density. The notebook below is simplification of the notebooks associated with the Dynamical Gravastar paper (which are at the URL: https://gitlab.com/stephenadler/Gravastar, and were written in Mathematica version 12.2), to the case of zero cosmological constant (lambda = 0) . So only the density jump mechanism is used in the demo notebook below . Including a cosmological constant makes a negligible correction to the plots from this demo notebook .

Extensive observations show that the universe contains a multitude of extremely compact objects, that are assumed to be mathematical black holes, as described in the monograph of Chandrasekhar [1]. Mathematical black holes are solutions of the vacuum Einstein field equations characterized by just two parameters, the mass M and the angular momentum per unit mass. But the interpretation of astrophysical observations in terms of idealized mathematical black holes has been questioned from various points of view.

Several authors, as reviewed by Cardoso and Pani [2], have proposed interior solutions for so-called "exotic compact objects" that appear black-hole like from the outside, but have no horizons and no interior singularity.
In particular, the gravastars proposed by Mazur and Mottola [3] are based on assuming a discontinuous pressure jump in the interior black hole equation of state, from a normal matter equation of state to the equation of state proposed by Gliner [4], in which the pressure p is minus the

density ρ.

The model developed in the following Mathematica notebook differs from that of Mazur and Mottola and the subsequent paper of Visser and Wiltshire [6] in several significant respects . First, we perform our entire analysis from the Tolman-Oppenheimer-Volkoff (TOV) equations for relativistic stellar structure (for a concise derivation see [7]).

Second, we note that the TOV equations require that the pressure p must be continuous, whereas the energy density ρ can have discontinuous jumps. So we implement the Gliner equation of state by a jump to negative energy density with positive pressure: at the pressure pjump the density jumps from a relativistic equation of state $\rho = 3\,p$ to an equation of state $\rho + p = \beta$, with $\beta << 1$ a parameter of the model. This of course violates the classical energy conditions, but from a semiclassical quantum matter point of view, the regularized energy density is known *not* to obey positivity conditions [5], [6].

Third, we avoid assuming designated radii at which transitions take place . In our model, transitions follow dynamically from the equations of motion and the assumed equations of state, hence the title of my paper "Dynamical Gravastars". And fourth, we smooth the jump in the equation of state by using a sigmoidal function with a very small switching width parameter eps = .001 in place of a Heaviside step function, so there are no exact discontinuities and accompanying surface densities to be considered. Thus we have a differential equation system that can be solved by the Mathematica integrator NDSolve, which is powerful general tool for solving one dimensional differential equation systems, such as arise from our assumptions when restricted to spherical symmetry .

Sample Mathematica notebooks for our model, for β parameter values $\beta = .1$, $\beta = .01$, and $\beta = .001$, can be downloaded at the URL given above. The programs begin with a list of numerical parameters, as shown for the three β values in Table I.

Notebook name	TOV.1	TOV.01	TOV.001
beta	.1	.01	.001
nuinit	−14.70	−21.255	−50.75
pjump	.7	.95	.98
lambda	$.3 \times 10^{-34}$	10^{-36}	$.4 \times 10^{-42}$, used 10^{-44}
rmax	10	60	80,000
rmin	10^{-7}	10^{-7}	10^{-7}
alpha0	−1	−1	−1
alpha1	3	3	3
kappa	8π	8π	8π
kappa2	4π	4π	4π
eps	.001	.001	.001

Numerical parameters for the Mathematica notebooks. We have eliminated the parameter lambda in the demo notebook that follows.

(Debug) In[ ]:=
```
beta = 10^(-2);
nuinit = -21.255;
pjump = .95;
rmax = 60.0;
rmin = 10^(-7);
alpha0 = -1;
alpha1 = 3;
kappa = 8 * Pi;
kappa2 = 4 * Pi;
eps = .001;
```

Following the initial parameter values list, there are three function definitions.The sigmoidal function is implemented by

(Debug) In[ ]:=
```
theta[x_] := 1 / (1 + Exp[-x / eps]);
```

The switch in the equation of state is implemented by the functions

(Debug) In[ ]:=
```
alphas[x_] := alpha0 * theta[x - pjump] + alpha1 * theta[pjump - x];
rho[x_] := alphas[x] * x + beta * theta[x - pjump];
```

In using these functions in the differential equation solver, x will always be the pressure $p[r]$.

After the function definitions, there follows setup of the system of differential equations to be solved. The variables nu[r], p[r], and em[r] correspond to $u(r)$, $p(r)$, and $m(r)$ in the TOV equations which are:

$$\frac{d\,m(r)}{d\,r} = 4\,\pi\,r^2\,\rho(r),$$

$$e^{-\lambda(r)} = 1 - \frac{2\,m(r)}{r}\;,$$

$$\frac{d\,\nu(r)}{d\,r} = \frac{N_\nu}{1 - 2\,m(r)/r},$$

$$N_\nu = (2/r^2)(m + 4\,\pi\,r^3\,p(r)),$$

$$\frac{d\,p(r)}{d\,r} = -\frac{\rho(r) + p(r)}{2}\,\frac{d\,\nu(r)}{d\,r}.$$

nu[r], p[r], and em[r] have respective initial values nuinit, 1, and 0 respectively, given in the first three lines within “system={....}”. The second three lines are the TOV differential equations, constructed using the functions defined in the preceding paragraph.

The integration range is taken to start from $r = 10^{-7}$ rather than $r = 0$ to avoid zero divides

(Debug) In[ ]:=
```
system = {
   nu[rmin] == nuinit,
   p[rmin] == 1,
   em[rmin] == 0,
  em'[r] == kappa2 * r^2 * rho[p[r]],
  nu'[r] == (2 / r^2) * (em[r] + kappa2 * r^3 * p[r]) / (1 - 2 * em[r] / r),
  p'[r] == - (1 / 2) * (p[r] + rho[p[r]]) * nu'[r]
 };
```

Next comes the command NDSolve for the system of equations.

(Debug) In[ ]:=
```
s = NDSolve[system, {nu, p, em}, {r, rmin, rmax}, PrecisionGoal → 13, AccuracyGoal → 13, MaxSteps → 10^10];
```

Next is extraction of the solution from the interpolating functions constructed by NDSolve,and computation of certain auxiliary quantities.

(Debug) In[ ]:=
```
pout[r_] := Evaluate[p[r] /. s];
nuout[r_] := Evaluate[nu[r] /. s];
nuprimeout[r_] := Evaluate[nu'[r] /. s];
emout[r_] := Evaluate[em[r] /. s];
denomout[r_] := (1 - 2 * emout[r] / r );
ennuout[r_] := (2 / r^2) * (emout[r] + kappa2 * r^3 * pout[r]);
rhoout[r_] := alphas[pout[r]] * pout[r] + beta * theta[pout[r] - pjump];
Mout[r_] := (1 - Exp[nuout[r]]) * r / 2;
```

Finally, comes printing and plotting of output from the solution

(Debug) In[ ]:=
```
Plot[rhoout[r] / pout[r], {r, rmin, rmax}, PlotRange → Automatic,
 AxesLabel -> {"r", "rho/p"}]
```

(Debug) Out[ ]=



(Debug) In[ ]:=
```
LogPlot[denomout[r], {r, rmin, rmax}, PlotRange → {10^(-4), 10^4}, AxesLabel -> {"r", "denom"}]
```

(Debug) Out[ ]=



(Debug) In[ ]:=
```
Plot[ Mout[r], {r, rmin, rmax}, PlotRange → {-1, 20}, AxesLabel -> {"r", "M[r]"}]
```

(Debug) Out[ ]=



(Debug) In[ ]:= `Plot[pout[r], {r, rmin, rmax}, PlotRange → {0, 1}, AxesLabel -> {"r", "p"}]`

(Debug) Out[ ]=



(Debug) In[ ]:= `Plot[rhoout[r], {r, rmin, rmax}, PlotRange → {-1, 2.8}, AxesLabel -> {"r", "rho"}]`

(Debug) Out[ ]=



(Debug) In[ ]:=
```
Plot[nuout[r], {r, rmin, rmax}, PlotRange → Automatic,
  AxesLabel -> {"r", "nu"}]
```

(Debug) Out[ ]=



(Debug) In[ ]:=
```
Plot[Exp[nuout[r]], {r, rmin, rmax}, PlotRange → Automatic,
  AxesLabel -> {"r", "g_00"}]
```

(Debug) Out[ ]=



(Debug) In[ ]:=

```
p1 = Plot[Exp[nuout[r]], {r, rmin, rmax}, PlotRange → Automatic,
  AxesLabel -> {"r", "exp[nu]"}, PlotStyle → Dotted]
```

(Debug) Out[ ]=



(Debug) In[ ]:=

```
p2 = Plot[1 - 33 / r, {r, rmin, rmax}, PlotRange → {0, .4},
  AxesLabel -> {"r", "1-33/r"}, PlotStyle → Dashed]
```

(Debug) Out[ ]=



(Debug) In[ ]:= **Show[p1, p2]**

(Debug) Out[ ]=



References

[1] S. Chandrasekhar, The Mathematical Theory of Black Holes (Clarendon Press, Oxford, 1983 (1992, 2000)), especially Chapters 2 and 6.

[2] V. Cardoso and P. Pani, Living Rev. Relativity 22, 4 (2019), arXiv : gr-qc/1904.05363.

[3] P. O. Mazur and E. Mottola, "Gravitational Condensate Stars", arXiv: gr-qc/0109035 (2001). See also Proc. Nat. Acad. Sci. **101**, 9545 (2004), arXiv: gr-qc/0407075.mazur

[4] E. B. Gliner, J . Exptl . Theoret . Phys . **49**, 542 (1965); translation in Sov . Phys . JETP **22**, 378 (1966).

[5] R. M. Wald, "General Relativity", The University of Chicago Press (1984), p .410.

[6] C. Barcelo and M. Visser, Int. J. Mod. Phys. D **11**, 1553 (2002), arXiv: gr-qc/0205066.

[7] Ya. B. Zeldovich and I. D. Novikov, *Stars and Relativity*, The University of Chicago Press (1971), pp .256 - 257.

INCLUSION OF COSMOLOGICAL CONSTANT

USED WEYL SCALING INVARIANT ACTION

$$S_{eff} = -\frac{\Lambda}{8\pi}\int d^4x \left({}^{(4)}g\right)^{1/2} (g_{00})^{-2}$$

$$\Rightarrow \quad p_{TOTAL} = p - \frac{\Lambda}{8\pi} e^{-2\nu(r)}$$

$$\rho_{TOTAL} = \rho - \frac{3\Lambda}{8\pi} e^{-2\nu(r)}$$

RELATIVISTIC ANALOG OF PARKER ISOTHERMAL PRESSURE DRIVEN WIND SHOWS THERE IS A VERY SMALL PARTICLE WIND PROPORTIONAL TO Λ, BIGGER THAN HAWKING FLUX BUT STILL NEGLIGIBLE FOR ASTROPHYSICAL APPLICATIONS

FURTHER GRAPHS + DISCUSSION arXiv: 2301.11821

EXTENSIONS, OPEN QUESTIONS

- AUTOMATE TUNING OF nainit
- EXPLORING PARAMETER SPACE
- ACCRETING GRAVASTARS
- AXIALLY SYMMETRIC, ROTATING GRAVASTARS
- EQUATION OF STATE VARIANTS
- STABILITY
- MODELS WITH SMOOTHED PRESSURE JUMP
- "RINGDOWNS" FROM GRAVASTAR MERGERS
- ANALYTIC APPROXIMATION FOR REGION OF NOMINAL HORIZON (I'm working on this with a student)
- ASTROPHYSICAL IMPLICATIONS OF ACCRETED MATTER LEAKING OUT g_{00} BECOMES SMALL → POSSIBLE LARGE TIME DELAYS

Galaxy formation?